
0.11ジュニア版     決定論と確率論

世の中の人々の経済的いとなみや物質のふるまいの中には、一見するとジグザグに見え

るものでも、決定した１つの数式で表せるふるまいと確率過程をともなった数式で表せる

ものがあることをここでは示しておこう。

決定論的なふるまい

プログラム

new;

cls;

/* Generating n data */

n=50;

x=zeros(n,1);

x[1]=0.2;

i=2;

do while i<=n;

   x[i]=4*x[i-1]*(1-x[i-1]);

   i=i+1;

endo;

/* Graph */

library pgraph;

pqgwin many;

   graphset;

      xlabel("t");

      ytics(0,1,0.1,0);

      xy(seqa(1,1,n),x);

   graphset;

      _plctrl=-1;

      xlabel("x(t)");

      ylabel("y(t+1)");

      xy(x[1:n-1],x[2:n]);



グラフ表示

上の１番目のグラフはｘ軸に時間ｔをとってデータをふるまいをプロットしたものである。

ジグザグの形をしていて、一見するとランダムにも見えるしサイクルがあるようにも見え

る。しかしながら、ｘ軸に当期の x(t)をとりｙ軸に次の期の x(t+1)をとってみると状況は一

変する。きれいな放物線を描いていることがわかる。

確率論的なふるまい

プログラム

new; cls;

n=10000;

xt=rndn(n,1);



x=cumsumc(xt);

library pgraph;

pqgwin many;

graphset;

   xlabel("t");

   xy(seqa(1,1,n),x);

graphset;

   _plctrl=-1;

   xlabel("x(t)");

   ylabel("y(t+1)");

   xy(x[1:n-1],x[2:n]);

graphset;

   xlabel("x(t+1)-x(t)");

   hist(x[2:n]-x[1:n-1],9);

グラフ表示

この例は、少しデータを多くしているが前の例と同じｘ軸に時間ｔをとっている。一見す

るとジグザグの中にも下降するトレンドがあるように思えてくる。しかしながら、この 1

番目のプロットは自分でプログラムを実行するとわかるように、その時そのときの乱数の

出具合によって下降トレンドにもなれば上方トレンドにもなるし、そうではない場合もあ

る。前の例と同じく、ｘ軸に当期の値 x(t)をｙ軸に次期の値 x(t+1)をとってプロットしてや

ると次のようなプロットになる。右上がりの直線となり、決定論的ではないにしても、何

らかの正の相関関係または回帰分析ができる関係にあるように見える。しかしながら、こ

の分析は実は正しくないことがすぐにわかる。



上の一番最後のヒストグラムのようにΔx(t)すなわち x(t+1)-x(t)の値をヒストグラムにして

やると美しいベルシェイプを描いていることがわかる。これはまさしく確率論そのものな

のである。

上の２つで見たことから少なくとも言えることは、何かのデータが与えられているなら

ばそのプロットだけから判断をしては誤りを犯すということである。少なくとも、当期と

次期の階差をとってみることが必要である。



パラメーターによるふるまいの変化

以下では、同じ決定論的ふるまいをするメカニズムであっても、そのとり得るパラメー

タの値によって時系列データが全くの違うものになることを示そう。

プログラム

new;

cls;

x1=0.1;

library pgraph;

graphset;

   pqgwin many;

   begwind;

   window(4,1,0);

      setwind(1);

         a=1;

         x=simplemay(x1,a);

         ytics(0,1,1,0);

         xy(seqa(1,1,rows(x)),x);

      setwind(2);

         a=1.1;

         x=simplemay(x1,a);

         ytics(0,1,1,0);

         xy(seqa(1,1,rows(x)),x);

      setwind(3);

         a=1.2;

         x=simplemay(x1,a);

         ytics(0,1,1,0);

         xy(seqa(1,1,rows(x)),x);

      setwind(4);

         a=1.99;

         x=simplemay(x1,a);

         ytics(0,1,1,0);

         xy(seqa(1,1,rows(x)),x);

      endwind;

   _plctrl=-1;

   xtics(0,1,0.5,0);



   ytics(0,1,1,0);

   xlabel("xt");

   ylabel("xt+1");

   xy(x[1:rows(x)-1],x[2:rows(x)]);

proc simplemay(x1,a);

   local n,x,i;

   n=100;

   x=zeros(n,1);

   x[1]=x1;

   i=1;

   do while i<=n-1;

      if x[i]<=0.5;

         x[i+1]=a*x[i];

      elseif x[i]>0.5;

         x[i+1]=a-a*x[i];

      else;

         errorlog "ERROR: Out of range.";

      endif;

      i=i+1;

   endo;

   retp(x);

endp;

グラフ表示

上のように、最初がａ＝１，それから順にａ＝1.1,ａ＝1.2,ａ＝1.99に増加させた場合に、

その時系列のふるまいが見かけ上、一定値→サイクル→複数サイクル→複雑なふるまいと



いうふうに変化をたどることを示している。しかしながら、ｘtとｘt+1の関係をグラフに

するとその背景には、次のような関係があることがわかる。

横軸にはｘtを縦軸にはｘt+1をとっている。その生成メカニズムのプログラム部分でもわ

かるように、端点が０と１で、頂点が(0.5,a/2)で、傾きがａと‐ａとなる三角形上の２辺の

上に軌跡があることがわかる。

Further Readings

特にマクロ経済学に興味のある読者は次の２冊を詳細に読むと得るところが大きいだろう

Day, R., 1994, Complex Economic Dynamics, Vol. 1: An Introduction to Dynamical   

   Systems and Market Mechanisms, MIT Press. ISBN: 0262041413

Day, R., 2000, Complex Economic Dynamics, Vol. 2: An Introduction to Macroeconomic

   Dynamics (Studies in Dynamical Economic Science), MIT Press. ISBN: 0262041723

